Relaxation method for solving Laplace's equation in two dimensions

Cell 1: Initialize the arrays

```
size = 50;  (* width and height of grid *)
vArray = Array[0.0 &, {size, size}]; (* array of potential values *)
fixedArray = Array[False &, {size, size}]; (* array of electrode sites *)
stepCount = 0;  (* number of relaxation steps so far *)
```

Cell 2: Create electrodes with fixed potentials (modify as desired)

```
vArray[[25, 20]] = 10;  (* 10 volts *)
fixedArray[[25, 20]] = True;  (* True means it's an electrode site *)
vArray[[25, 30]] = -10;
fixedArray[[25, 30]] = True;
```

Cell 3: Create lists of potential and field values for plotting, then plot them.
(You can choose either a density plot or contour plot by commenting-out one or the other.)

```
v = Flatten[Table[{x, y, vArray[[x, y]]}, {x, 1, size}, {y, 1, size}], 1];
e = Flatten[Table[{{x, y}, {-(vArray[[x + 1, y]] - vArray[[x - 1, y]]) / 2,
       -(vArray[[x, y + 1]] - vArray[[x, y - 1]]) / 2}},
    {x, 2, size - 1}, {y, 2, size - 1}], 1];
(*vPlot = ListContourPlot[v,ColorFunction→(Blend[{Green,White,Magenta},#]&),
       Contours→19,PlotRange→All]; *)
vPlot = ListDensityPlot[v, ColorFunction → (Blend[{Green, White, Magenta}, #] &),
       PlotRange → All, InterpolationOrder → 0];
ePlot = ListVectorPlot[e];
Show[vPlot, ePlot]
```

Cell 4: Define a function to execute n relaxation steps

```
relax[n_] := (For[step = 1, step ≤ n, step++,
                stepCount++;
                For[x = 2, x < size, x++,
                 For[y = 2, y < size, y++,
                  If[fixedArray[[x, y]], ,
                   vArray[[x, y]] = (vArray[[x - 1, y]] + vArray[[x + 1, y]] +
                                     vArray[[x, y - 1]] + vArray[[x, y + 1]]) / 4;]]]];
              stepCount)  (* show the number of steps so far *)
```

Cell 5: Carry out the desired number of relaxation steps

```
relax[1]
```